Problem 2.10

Problem 2.10. Use a computer to produce a table and graph, like those in this section, for the case where one Einstein solid contains 200 oscillators, the other contains 100 oscillators, and there are 100 units of energy in total. What is the most probable macrostate, and what is its probability? What is the least probable macrostate, and what is its probability?

Einstein solid with 'n' oscillators and 'q' energy units:

In[119]:=

$$\Omega[n\_, q\_] := \frac{(q + n - 1)!}{q!\,(n - 1)!}$$

Reproduce the table in Figure 2.4.

Systems A and B

In[120]:=

```
nA = 200; nB = 100; qTotal = 100;
```

In[121]:=

```
ntotal = Ω[nA + nB, qTotal]; N[ntotal]
```

Out[121]=

1.68139×10^{96}

In[122]:=

```
data = Table[{qA, ΩA = Ω[nA, qA], qB = qTotal - qA,
    ΩB = Ω[nB, qB], ΩA * ΩB, N[ΩA * ΩB / ntotal] }, {qA, 0, qTotal}];
```

In[123]:=

```
data = Table[{qA, ΩA = N[Ω[nA, qA]], qB = qTotal - qA,
    ΩB = N[Ω[nB, qB]], ΩA * ΩB, N[ΩA * ΩB / ntotal] }, {qA, 0, qTotal}];
```

In[124]:=

```
TableForm[data⟦Range[10]⟧,
 TableHeadings → {None, {"qA", "ΩA", "qB", "ΩB", "Ωtotal", "Probability"}}]
```

Out[124]//TableForm=

qA	ΩA	qB	ΩB	Ωtotal	Probability
0	1.	100	4.52743×10^{58}	4.52743×10^{58}	2.69267×10^{-38}
1	200.	99	2.27509×10^{58}	4.55018×10^{60}	2.7062×10^{-36}
2	20 100.	98	1.13754×10^{58}	2.28646×10^{62}	1.35986×10^{-34}
3	1.3534×10^{6}	97	5.65885×10^{57}	7.65869×10^{63}	4.55497×10^{-33}
4	6.86851×10^{7}	96	2.80055×10^{57}	1.92356×10^{65}	1.14403×10^{-31}
5	2.80235×10^{9}	95	1.37873×10^{57}	3.86369×10^{66}	2.29792×10^{-30}
6	9.5747×10^{10}	94	6.75153×10^{56}	6.46439×10^{67}	3.84467×10^{-29}
7	2.8177×10^{12}	93	3.28831×10^{56}	9.26546×10^{68}	5.51059×10^{-28}
8	7.29079×10^{13}	92	1.59278×10^{56}	1.16126×10^{70}	6.90654×10^{-27}
9	1.68498×10^{15}	91	7.67201×10^{55}	1.29272×10^{71}	7.68839×10^{-26}

In[125]:=

```
ListPlot[data[[All, {1, 6}]], LabelStyle → Larger,
 AxesLabel → {"qA", "Probability"}, ImageSize → Scaled[0.8], PlotRange → All]
```

Out[125]=



Most-probable macrostate

The most probable state is qA = 67.

In[126]:=

```
max = MaximalBy[data, #[[6]] &]
```

Out[126]=

$\{\{67, 8.91338\times10^{63}, 33, 1.3799\times10^{31}, 1.22996\times10^{95}, 0.0731514\}\}$

Least-probable macrostate

The least probable state is qA = 0.

In[127]:=

```
min = MinimalBy[data, #[[6]] &]
```

Out[127]=

$\{\{0, 1., 100, 4.52743\times10^{58}, 4.52743\times10^{58}, 2.69267\times10^{-38}\}\}$